

## 3.6 Koordinatensysteme

Sehr oft werden auf dem Bildschirm Graphen gezeichnet, deren Punkte mit Hilfe von Zuordnungsvorschriften berechnet werden. Meist handelt es sich dabei um *Funktionsgraphen*, wie sie aus dem Mathematikunterricht gut bekannt sind. Dabei treten immer wieder die gleichen Probleme auf:


- Punkte des „mathematischen“ Koordinatensystems müssen auf „Bildschirmkoordinaten“ umgerechnet werden.
- Zahlenwerte müssen geändert, ggf. vom Benutzer eingegeben werden. Dabei sollten eventuelle Fehler abgefangen werden.
- die geänderten Graphen müssen am Bildschirm dargestellt werden.

Wir wollen diese Probleme nacheinander lösen und dann in einigen Anwendungen einsetzen.

### 3.6.1 Koordinatentransformationen

Mathematische und Bildschirmkoordinaten unterscheiden sich in drei Punkten:

- Der *Ursprung*, also der Punkt  $(0|0)$ , liegt beim mathematische Koordinatensystem meist in der Mitte des beschriebenen Blattes; auf dem Bildschirm befindet er sich in der linken oberen Ecke der Canvas.
- Die *y-Achse* ist bei mathematischen Koordinaten nach oben gerichtet, auf dem Bildschirm aber nach unten.
- Die *Längeneinheit* ist im mathematischen Koordinatensystem durch das Intervall von Null bis Eins gegeben, auf dem Bildschirm beträgt sie ein Pixel (einen Bildpunkt).



Trotz aller Unterschiede sollen die erzeugten Graphen in beiden Koordinatensystemen ähnlich sein, d. h. sich entsprechende Strecken sollten im gleichen Verhältnis zueinander stehen. Dieses Ziel nutzen wir aus. Wir fordern:

*Die Strecke  $x$  vom Nullpunkt des mathematischen Koordinatensystems zur  $x$ -Koordinate eines Punktes verhält sich zur Längeneinheit 1 dieser Achse genauso, wie die entsprechenden Strecken auf dem Bildschirm. Für die  $y$ -Koordinaten gilt Entsprechendes.*

Bezeichnen wir (wie in der Zeichnung angegeben) die Bildschirmkoordinaten des Ursprungs des auf den Bildschirm abgebildeten mathematischen Koordinatensystems als  $x_0$  bzw.  $y_0$  und die Bildschirmkoordinaten der Punkte, die einer Längeneinheit auf den Achsen entsprechen, mit  $xEins$  bzw.  $yEins$ , dann erhält man die Bildschirmkoordinaten  $x_B$  bzw.  $y_B$  eines Punktes  $P(x|y)$ , indem man die entsprechenden Verhältnisgleichungen aufstellt:

$$\frac{x}{1} = \frac{x_B - x_0}{xEins - x_0} \qquad \frac{y}{1} = \frac{y_B - y_0}{yEins - y_0}$$

Diese Beziehungen können wir je nach Bedarf entweder nach  $x$  und  $y$  auflösen (wenn wir die Bildschirmkoordinaten kennen und die entsprechenden mathematischen Koordinaten berechnen wollen), oder nach  $x_B$  und  $y_B$ , wenn wir aus den mathematischen die Bildschirmkoordinaten berechnen. Es ergibt sich:

$$\begin{aligned} x &= \frac{x_B - x_0}{xEins - x_0} & x_B &= x \cdot (xEins - x_0) + x_0 \\ y &= \frac{y_B - y_0}{yEins - y_0} & y_B &= y \cdot (yEins - y_0) + y_0 \end{aligned}$$

### 3.6.2 Zahleneingabe mit einem Editierfeld

Mit Hilfe von *Editierfeldern*, also Komponenten vom Typ *tEdit*, können einzeilige Texte auf dem Bildschirm dargestellt und vom Benutzer verändert werden, wenn das vorgesehen ist (die Eigenschaft *ReadOnly* muss dann auf *false* stehen). Der aktuelle Text des Editierfeldes ist in der Eigenschaft *Text* enthalten. Heißt also z. B. ein Editierfeld *FeldyEins*, dann steht der aktuelle Wert als *FeldyEins.Text* zur Verfügung. In dieser Variablen können wir aber nur Zeichenketten speichern, d. h. wir müssen unsere Zahlen bei Bedarf in Zeichenketten verwandeln – und umgekehrt. Zu diesem Zweck stehen verschiedene *Typwandlungsfunktionen* zur Verfügung:

- *IntToStr(zahl)* wandelt eine ganze Zahl in einen String.

- *StrToInt(string)* bestimmt die in einem String enthaltene ganze Zahl.
- *StrToFloat(string)* bestimmt die in einem String enthaltene Gleitpunktzahl.
- *FloatToStrF(gpzahl,ffFixed,laenge,stellen)* verwandelt die Gleitpunktzahl *gpzahl* in einen String mit fester Stellenzahl (daher die Delphi-Konstante *ffFixed*), dessen Länge und Nachkommastellen als Nächstes angegeben werden.

Üblicherweise wird beim Programmwurf dem benutzten Editierfeld ein Anfangswert mit Hilfe des Objektinspektors zugewiesen. Wird dieser Wert während des Programmlaufs geändert, dann tritt das *OnChange*-Ereignis des Editierfeldes ein. In der zugeordneten Ereignisbehandlungsprozedur wird dann die neue Zahl bestimmt, z. B. durch


```
xEins := StrToInt(FeldxEins.Text);
```

Was passiert aber, wenn der Benutzer einen Fehler macht, also eine Zeichenkette eingibt, die gar keine Zahl enthält. (Meist passiert das, wenn man die alte Zahl löscht. Dann befindet sich kurzzeitig nur ein leerer String im Eingabefeld!) Unser Programm würde mit einer Fehlermeldung abrechen – und das wollen wir natürlich verhindern.

Wir benutzen deshalb einen *try...except..*Block, in dem wir versuchen, die in einem String enthaltene Zahl zu bestimmen. Tritt dabei ein Fehler (eine *Exception*) auf, dann werden die im *except*-Teil vorgesehenen Anweisungen ausgeführt. (*Vorsicht*: das geschieht nur, wenn im Delphisystem unter dem Menüpunkt „Tools – Umgebungsoptionen“ der Auswahlpunkt „bei Exceptions anhalten“ nicht angewählt ist!)

Das (unvollständige, aber für unsere Zwecke ausreichende) Syntaxdiagramm dafür lautet:

#### try...except-Anweisung:



Damit lautet die vollständige Ereignisbehandlungsroutine:

```

procedure TForm1.FeldxEinsChange(Sender: TObject);
begin
try
  xEins := StrToInt(FeldxEins.text)
except
  FeldxEins.Text := IntToStr(100);
  xEins := 100
end
end;
```

### 3.6.3 Zahleneingabe mit einem Schieberegler

Soll sich ein Wert nur in einem vorgegebenen Zahlenbereich ändern können, dann kann diese Änderung mit Hilfe einer *Scrollbar*-Komponenten erfolgen, die wir z. B. von den Rändern eines Windows-Fensters her kennen, in dem sich mehr befindet, als bei der momentanen Fenstergröße dargestellt werden kann. Komfortabel wird so eine Zahleneingabe, wenn der eingestellte Wert direkt in einem Editierfeld dargestellt wird. (Beschränken wir uns auf die reine Anzeige, dann sollte dessen *ReadOnly*-Eigenschaft auf *TRUE* gesetzt werden.) Da die so bestimmten Zahlen im Programm berechnet werden, kann auch keine fehlerhafte Eingabe erfolgen, so dass wir uns *einen try...except...Block* ersparen.



Der Wert einer Scrollbar-Komponenten kann sich standardmäßig zwischen den Werten 0 und 100 bewegen. (Das kann im Objektinspektor auch abgeändert werden.) Die momentane Position des Schiebereglers ist in seiner *Position*-Eigenschaft gespeichert. Aus deren Wert und dem Maximalwert kann dann leicht der gewünschte Wert abgeleitet werden. Soll sich z. B. eine Variable *x* mit Hilfe einer Scrollbar *Scrollx* zwischen den Werten 1 und 2,5 ändern, dann erhalten wir:

```
x := 1 + 1.5*Scrollx.Position/100;
```


Bewegt man den Schieberegler mit der Maus, dann wird ein *OnChange*-Ereignis ausgelöst, auf das mit einer Ereignisbehandlungsroutine reagiert wird. In dieser wird eine Zahl *x* bestimmt und im Editierfeld dargestellt:

```
procedure TForm1.ScrollxChange(Sender: TObject);
begin
  x := 1 + 1.5*Scrollx.Position/100;
  Zeigex.Text := FloatToStrF(x,ffFixed,6,2);
end;
```

### 3.6.4 Funktionsgraphen

Wir wollen unsere Kenntnisse jetzt auf ein Standardgebiet der Schulmathematik anwenden: das Zeichnen von Funktionsgraphen von Parabeln der Form  $y = a \cdot x^2 + b \cdot x + c$ . Dazu erzeugen wir auf unserer Form eine *Panel*-Komponente, die drei Schieberegler mit Anzeigefeldern aufnimmt, mit deren Hilfe wir die Werte der Parameter *a*, *b* und *c* zwischen den Werten  $-5$  und  $5$  ändern können. Setzen wir die *Align*-Eigenschaft des Panels auf *allLeft*, dann „klebt“ dieses Element am linken Rand des Fensters. Den Rest der Fensterfläche soll eine *PaintBox*-Komponente einnehmen, deren *Canvas* die Funktionsgraphen anzeigt. Ihre

*Align*-Eigenschaft wird deshalb auf *aClient* gesetzt. Eine *Timer*-Komponente steuert das Zeichnen der Graphen, weil diese zwischen den Zeichenvorgängen die Kontrolle an Windows zurückgibt, so dass auch andere Elemente der Windows-Oberfläche mal drankommen können.



Immer wenn ein Parameter oder das Fenster geändert wurde, soll ein neuer Graph gezeichnet werden. Dazu schreiben wir ein Unterprogramm *FangeNeuAn*, in dem die Anfangswerte gesetzt und die Achsen des Koordinatensystems neu gezeichnet werden.

```

procedure FangeNeuAn;
begin
  a := -5+10*Form1.ScrollA.Position/100;
  b := -5+10*Form1.ScrollB.Position/100;
  c := -5+10*Form1.ScrollC.Position/100;
  with Form1.Bild do
    begin
      x0 := width DIV 2;
      y0 := height DIV 2;
      xEins := x0 + x0 DIV 3;
      yEins := y0 - y0 DIV 3;
    
```

Parameterwerte  
aus den Schiebe-  
reglerstellungen  
bestimmen

Ursprung und  
Längeneinheiten aus der  
Fenstergröße berechnen

```

with canvas do
  begin
    pen.color := clBlack;
    pen.style := psSolid;
    MoveTo(0,y0);      LineTo(2*x0,y0);
    MoveTo(x0,0);      LineTo(x0,2*y0);
    MoveTo(xEins,y0+5); LineTo(xEins,y0-5);
    MoveTo(x0-4,yEins); LineTo(x0+5,yEins);
  end
end;
xB := 0;
end;

```

Koordinatensystem  
zeichnen

Dieses Unterprogramm wird in den entsprechenden Ereignisbehandlungsroutinen aufgerufen, nachdem die neuen Werte bestimmt sind. In der Timer-Prozedur wird der Funktionswert in Abhängigkeit von der momentanen Bildschirmkoordinate  $x_B$  berechnet. Danach wird eine entsprechende Linie gezeichnet und der Wert von  $x_B$  erhöht.

```

procedure TForm1.Timer1Timer(Sender: TObject);
begin
  if xB > 2*x0 then exit;
  x := (xb-x0)/(xEins-x0);
  y := a*x*x + b*x + c;
  yB := Round(y*(yEins-y0)+y0);
  if xB = 0
  then Bild.Canvas.MoveTo(xB,yB)
  else Bild.Canvas.LineTo(xB,yB);
  xB := xB + 1;
end;

```

Ende, falls am  
rechten Rand

mathematische  
Koordinaten

Umrechnung auf  
Bildschirmkoordinaten

... und dann zum  
nächsten Punkt

zuerst zum  
Anfangspunkt,  
danach Linien

Zuletzt wird ein Button *Neu* eingeführt, dessen Ereignisbehandlungsroutine das gesamte Bild löscht.

```

procedure TForm1.NeuClick(Sender: TObject);
begin
  with Form1.Bild do
  begin
    canvas.brush.color := clWhite;
    canvas.Brush.style := bsSolid;
    canvas.Rectangle(0,0,width,height)
  end;
  FangeNeuAn
end;

```

den Brush auf weißen  
Hintergrund einstellen...

... und gesamte  
Canvas löschen

Da alle Variable in unterschiedlichen Unterprogrammen benötigt werden, müssen sie *global* vereinbart werden - entweder schon im Interface-Teil oder direkt am Anfang des Implementation-Teils; in jedem Fall *bevor* sie das erste Mal benutzt werden.

```
var
  a,b,c,x,y           : Double;
  x0,y0,xEins,yEins,xB,yB: integer;
```

### 3.6.5 Aufgaben

1. Schreiben Sie Programme, um die **Graphen**
  - a: gebrochen rationaler Funktionen der Form  $y = \frac{a \cdot x + b}{c \cdot x + d}$  auf dem Bildschirm darzustellen.
  - b: von Potenzfunktionen der Form  $y = a \cdot x^n$  darzustellen.
  - c: von Exponentialfunktionen der Form  $y = a \cdot e^x \cdot (b \cdot x^n + c \cdot x + d)$  darzustellen.
  - d: von Logarithmusfunktionen darzustellen.

Benutzen Sie bei Bedarf *try...except*...Blöcke zur Berechnung der Funktionswerte.
2. Verändern Sie das Programm zum **Graphenzeichen** so, dass
  - a: die Skalen auf den Achsen verändert werden können.
  - b: der dargestellte Bereich verschoben werden kann.
  - c: durch Auswahl eines rechteckigen Bereichs mit der Maus automatisch gezoomt werden kann.
3. Schreiben Sie ein Programm, in dem eine **Wertetabelle** ausgefüllt werden kann, indem die Wertepaare in zwei Editierfelder eingegeben und auf Knopfdruck in ein Feld (z. B. *tFeld=array[1..20,1..2] of Single*) übernommen werden. Das Feld wird am Bildschirm dargestellt, indem auf der Canvas in Texte verwandelte Zahlen geschrieben werden. Nach Abschluss der Eingabe soll ein entsprechender Graph gezeichnet werden.

4. a: Die Eckpunkte eines **Dreiecks** sollen in der Koordinatendarstellung erfragt werden: z. B. als  $A(-5|-3)$ ,  $B(6|1)$  und  $C(2|5)$ . Dazu benötigt man sechs Editierfelder. Bei Bedarf kann die Zahleneingabe auch mit Hilfe von Schieberegler erfolgen.  
b: Schon bei der Zahleneingabe sollen die Seitenlängen, die Winkel und die Fläche des Dreiecks berechnet und in Editierfeldern oder Label-Komponenten angezeigt werden. Mit Hilfe von *try...except...* Blöcken sollen Fehler abgefangen werden.  
c: Das Dreieck soll am Bildschirm dargestellt werden.  
d: Auf entsprechende Knopfdrücke sollen Umkreis, Inkreis, Mittelsenkrechte, Seitenhalbierende, ... gezeichnet werden.
5. a: Der **Thaleskreis** soll am Bildschirm dargestellt werden. Mit Hilfe eines Schiebereglers soll der Eckpunkt des Dreiecks auf dem Kreis verschoben werden können.  
b: Über den Seiten des Dreiecks sollen die entsprechenden Quadrate gezeichnet und ihre Flächen berechnet und angezeigt werden. Die Summe der Flächen der beiden kleineren Quadrate wird ebenfalls angezeigt. (*Satz des Pythagoras*)  
c: Über der Höhe des Dreiecks sollen das Quadrat, über einem Höhenabschnitt das Rechteck mit der Seitenlänge des zweiten Höhenabschnitts gezeichnet und die Flächen berechnet und angezeigt werden. (*Höhensatz des Euklid*)  
d: Über einer Seite des Dreiecks soll das Quadrat, über der Hypotenuse das Rechteck mit der Seitenlänge des entsprechenden Höhenabschnitts gezeichnet und die Flächen berechnet und angezeigt werden. (*Kathetensatz des Euklid*).
6. a: Ein Kreis wird gezeichnet und in ihm eine Sehne. Die Enden der Sehne werden mit dem Kreismittelpunkt verbunden. Die Lage der Sehne kann mit Hilfe eines Schiebereglers verändert werden.  
b: Ein Punkt auf dem Kreisbogen über Sehne soll mit Hilfe eines Schiebereglers verschoben werden können. Der Punkt wird mit den Enden Sehne verbunden.  
c: Die Winkel in den entstandenen Dreiecken am Kreisbogen und am Mittelpunkt des Kreises sollen berechnet und angezeigt werden. (*Satz über die Peripheriewinkel*)

## 3.7 Simulationen und chaotisches Verhalten

### 3.7.1 Die Chaosparabel

Simuliert man mit Computern das Verhalten realer Systeme, dann geschieht das oft, indem man mithilfe bekannter Gesetzmäßigkeiten aus dem derzeitigen Zustand den darauf folgenden berechnet, und aus diesem wiederum einen späteren, usw. Man steckt also die Zwischenergebnisse wieder als Eingabewerte in eine Gleichung (oder ein Gleichungssystem), und berechnet daraus neue (Zwischen-)Ergebnisse. Eine typische Anwendung ist das Wetter von nächster Woche, für das aus den heute gemessenen Temperaturen, Luftdrücken, Niederschlägen, ... das Wetter von morgen berechnet wird, und daraus das von übermorgen usw. Da Wettervorhersagen ziemlich ungenau sind, muss man sich fragen, woher diese Unsicherheit kommt, wenn doch genaue Messwerte und schnelle Supercomputer bereitstehen.

Bezeichnen wir den betrachteten Zustand zu einem Zeitpunkt als  $x_t$  und den darauffolgenden als  $x_{t+1}$ , dann berechnen wir  $x_{t+1}$  aus  $x_t$  mit Hilfe der Funktion  $f$ :

$$x_{t+1} = f(x_t)$$

Man spricht von einer *Iteration*. Eine sehr einfache solche Funktion wird durch die *logistische Gleichung* beschrieben, eine einfache quadratische Funktion:


$$x_{t+1} = a \cdot x_t \cdot (1 - x_t)$$

(Ein Beispiel für die Herleitung einer Beziehung dieser Art folgt im nächsten Kapitel.)


Zeichnen wir den zugehörigen Funktionsgraphen, dann erhalten wir eine nach unten geöffnete Parabel, die die x-Achse in den Punkten 0 und 1 schneidet. Der Parameter  $a$  bestimmt die Steilheit der Parabel und kann durch eine Scrollbar verändert werden, die am rechten Rand des Fensters klebt. Den Startwert  $x_1$  können wir ebenfalls mit Hilfe einer Scrollbar-Komponenten bestimmen. (**Achtung:** Die *Align*-Eigenschaft dieser Komponenten erscheint bei älteren Delphi-Versionen nicht im Objekt-Inspektor. Sie kann aber z. B. in der *OnCreate*-Methode des erzeugten Fensters auf einen geeigneten Wert gesetzt werden; z. B. *ScrollbarX.Align := alBottom*.) Der erste berechnete Funktionswert wird dann durch eine senkrechte Linie von diesem Wert zur Parabel angezeigt.

Es fehlt jetzt nur noch eine grafische Veranschaulichung des Wieder-Hineinsteckens des Funktionswerts in die Berechnungsvorschrift. Das Vertauschen der Rollen von x- und y-Wert entspricht in der Grafik einer Vertauschung der beiden Koordinatenachsen, und das kann durch eine Spiegelung an der Geraden  $y=x$ , also der Hauptdiagonalen, geschehen. Wir zeichnen deshalb eine horizontale Linie vom erreichten Punkt der Parabel zu dieser Diagonalen, erhalten damit den neuen x-Wert und bestimmen den zugehörigen Funktionswert, indem wir wiederum senkrecht nach oben (oder unten) zur Parabel zeichnen.

Dieser Prozess wird dann laufend wiederholt. Das Ergebnis ist eine Art Treppenkurve, die unabhängig vom Startwert immer am Schnittpunkt von Parabel und Diagonale endet – wenn denn der Parameter  $a$  nicht zu groß und damit die Parabel nicht zu steil wird.



Vergrößern wir jetzt den Wert von  $a$ , dann verläuft die Treppenkurve (meist spiralförmig) immer noch auf diesen Schnittpunkt zu, benötigt aber deutlich mehr Rechenschritte bis zum Ende. Bei weiterer Erhöhung läuft die Kurve in periodische Zustände hinein, bei denen sie ohne weitere Annäherung an den Schnittpunkt immer wieder den gleichen Kurvenzug (oft ein Rechteck) beschreibt. Bei noch größeren Werten von  $a$  (und noch steilerer Parabel) erreicht die Treppenkurve den Schnittpunkt nicht mehr. Statt dessen springt sie völlig unregelmäßig mal nahe an den Schnittpunkt heran und dann wieder weit von ihm weg. Leichte Veränderungen des Anfangswertes führen zu völlig anderen Verläufen der Iteration. Der Kurvenverlauf wird *chaotisch*.



Was bedeutet das nun?


Interpretieren wir den Startwert  $x_1$  als „das Wetter von heute“, dann bedeutet das Ergebnis des 30. Iterationsschritts „das Wetter in einem Monat“. Dieses Ergebnis ändert sich aber schlagartig (praktisch im gesamten möglichen Wertebereich), wenn man den Startwert nur minimal ändert. Bestehen also nur geringste Unsicherheiten z. B. über die Temperaturverteilung heute (und die bestehen immer), dann ist das Wetter schon über eine geringe Zeitspanne praktisch unvorhersagbar! Anschaulich spricht man davon, dass „ein Schmetterlings-Flügel Schlag in Südamerika nach einiger Zeit einen Taifun in China“ verursachen kann.

Den Zusammenhang zwischen der Änderung des Startwertes und dem Ergebnis nach einer festen Zahl von Iterationsschritten lässt sich grafisch leicht darstellen, wenn man über dem Startwert den nach  $n$  Iterationen berechneten Endwert aufträgt.

Iteriert man nur dreimal („Dreitage-wetter“), dann ergibt sich eine „glatte“ Kurve: benachbarte Startwerte führen zu ähnlichen Ergebnissen.

Bei  $n=11$  („11-Tage-Wettervorhersage“) ergibt sich zwar in der Mitte des Intervalls noch eine recht glatte Kurve, an den Seiten springen die Ergebnisse aber schon sehr willkürlich.

Für  $n=31$  („die Monats-Wettervorhersage“) springen die Ergebnisse praktisch beliebig zwischen benachbarten Werten. Eine Vorhersage des Ergebnisses ist unmöglich.



### 3.7.2 Populationsentwicklungen


Ein weiterer Anwendungsfall für Iterationen ist die Simulation der Bevölkerungsentwicklung in einem abgeschlossenen Lebensraum, in dem nur gleichartige Individuen leben. (Als Beispiel kann eine Bakterienkultur in einer Petrischale dienen.). Gilt für diesen Raum eine *Geburtenrate*  $g$  und eine *Sterberate*  $s$  der Individuen, dann erhält man aus einfachen Überlegungen die Bevölkerungszahl  $N_{t+1}$  einer Generation aus der vorhergehenden  $N_t$  zu

$$N_{t+1} = N_t + (g - s) \cdot N_t$$

Ist die Sterberate konstant und größer als die (konstante) Geburtenrate, dann stirbt die Population aus, ist sie kleiner, dann wächst die Bevölkerung exponentiell. (**Modell 1: exponentielles Wachstum**)

Man kann die Simulation aber noch verbessern, indem man die Sterberate von der Bevölkerungszahl abhängig macht. Ist die Sterberate proportional zu  $N$  (z. B. weil sich die Ernährungslage bei wachsender Bevölkerungszahl verschlechtert), dann haben wir das **Modell 2: Wachstum mit Behinderung**.

Ist die Sterberate proportional zur Gesamtzahl der Individuen, die bisher im Lebensraum gelebt haben (z. B. weil diese die Umwelt vergiften oder Ressourcen verbraucht haben), dann spricht man vom **Modell 3: Wachstum mit Selbstvergiftung**.



Wir können das ausgewählte Modell durch einen Satz von *Radiobuttons* bestimmen, und die Geburten- und Sterberate mit *Schiebereglern* einstellen. Veränderte Skalenfaktoren für die Achsen und eine Verschiebung des Anfangswert gestatten es dann, beliebige Ausschnitte der gezeichneten Kurve zu betrachten.

Überlassen wir das Bestimmen der Anfangswerte und das Zeichnen des Koordinatensystems wieder einem Unterprogramm *FangeNeuAn*, dann ändert sich beim Drücken z. B. des Radiobuttons *RBExp*, mit dem das Modell Nr.1 ausgewählt wird, nur die entsprechende Ganzzahlvariable *Modell*. Danach geht das Zeichnen neu los.

```

procedure TForm1.RBexpClick(Sender: TObject);
begin
  Modell := 1;
  FangeNeuAn;
end;

```

Modell wählen ...

... und Neuanfang

Die schon bekannte Timer-Komponente steuert das Zeichnen, indem sie entsprechend dem Wert der Variablen *Generation* (die der *x*-Koordinaten entspricht) den Graph fortsetzt. Da bei exponentiellem Wachstum Zahlenüberläufe möglich sind, werden die Funktionswerte wieder in einem *try...except...Block* berechnet.

```

procedure TForm1.Timer1Timer(Sender: TObject);
begin
  try
    case Modell of
      1: n := n + (g-s)*n;
      2: n := n + (g-s*n)*n;
      3: n := n + (g-s*nSumme)*n;
    end
  except
    n := 0;
    Meldungen.Caption := 'Zahlenüberlauf!';
  end;

  if n < 0 then n := 0;
  if (n > Bild.height*dy*1000) then exit;
  if Generation > Bild.width*dx then exit;
  nSumme := nSumme + n;

  if (dx <> 0) and (dy <> 0) then
    Bild.Canvas.LineTo(Round(Generation/dx+x0),
    y0-Round(n/dy));
  Generation := Generation + 1;
end;

```

je nach Modell  
Populationszahl  
berechnen

Katastrophen melden

nur sinnvolle Werte  
berechnen

Graph fortsetzen

Die Geburtenrate z. B. wird in der schon bekannten Art mit einem Schieberegler *ScrollBarG* eingestellt und in einem Editierfeld *GebRate* angezeigt.

```

procedure TForm1.ScrollBarGChange(Sender: TObject);
begin
  GebRate.Text :=
    FloatToStrF(2*Form1.ScrollbarG.Position/1000,ffFixed,6,4);
  g := StrToFloat(GebRate.Text);
  FangeNeuAn
end;

```

Der Skalenfaktor *dy* der y-Achse wird dagegen direkt in ein Editierfeld *SkalaY* eingegeben. Deshalb müssen eventuelle Eingabefehler wieder in einem *try...except...Block* abgefangen werden.


```

procedure TForm1.skalayChange(Sender: TObject);
begin
  try
    dy := StrToFloat(SkalaY.text)
  except
    SkalaY.Text := IntToStr(1);
    dy := 1
  end;
  FangeNeuAn
end;

```

Zahl ermitteln


Interessant ist das Modell2 („Wachstum mit Behinderung“), bei dem sich nach einigen Generationen ein Gleichgewichtszustand einstellt, der sich in einer konstanten Bevölkerungszahl bemerkbar macht.



### 3.7.3 Chaotische Bevölkerungsdynamik

Wir wollen uns jetzt für die Frage interessieren, wie die stabile Bevölkerung nach Modell 2, also die nach einigen Generationen erreichte Bevölkerungszahl, bei fester Sterberate von der Geburtenrate abhängt. Dazu variieren wir die Geburtenrate in einem vorgegebenen Bereich, berechnen mit immer gleichem Startwert einige Iterationswerte (z. B. 50 Generationen) und tragen die nächsten (z. B. 200) Generationen im Diagramm über der Geburtenrate auf. Wir erwarten natürlich, dass sich (zumindest fast) immer der gleiche Wert ergibt, so dass diese Werte alle dicht um oder auf einem Punkt liegen.

Anfangs erhalten wir auch so ein Ergebnis, und da die stabile Bevölkerungszahl mit steigender Geburtenrate zunimmt, ergibt sich eine ansteigende Gerade. Aber dann passiert etwas!



Mit weiter steigender Geburtenrate „springt“ die Bevölkerungszahl plötzlich zwischen zwei Werten, und diese Äste gabeln später wieder und wieder. Man spricht von *Bifurkationsprozessen*. Im rechten Teil des Diagramms schwankt die Bevölkerungszahl von Generation zu Generation zwischen beliebigen Werten, und nahe beieinanderliegende Geburtenraten führen zu unvorhersagbaren Sprüngen im Ergebnis. Das Verhalten wird chaotisch.

### 3.7.4 Aufgaben

1. Schreiben Sie ein Programm, das
  - a: die **Chaosparabel** entsprechend den Bildern in Kapitel 3.7.1 erzeugt.
  - b: die Ergebnisse einer einzustellenden Zahl von Iterationen der **logistischen Gleichung** wie in Kapitel 3.7.1 angegeben darstellt.
  - c: die **Entwicklung der Populationen** nach den drei Modellen wie in Kapitel 3.7.2 angegeben darstellt.
  - d: das **Bifurkationsdiagramm** aus Kapitel 3.7.3 erzeugt. Die Skalierung der y-Achse und das Intervall für die Geburtenrate sollen veränderbar sein.
  - e: das Bifurkationsdiagramm aus Kapitel 3.7.3 erzeugt. Mit Hilfe der Maus sollen **rechteckige Ausschnitte** aus dem Diagramm markiert werden können. Diese werden dann in einem nächsten Schritt vergrößert dargestellt.
  
2. Leben Hasen und Füchse in einem gemeinsamen Lebensraum, dann haben wir es mit einem **Räuber-Beute-System** zu tun: Füchse fressen Hasen und vermehren sich dabei gut. Leider nimmt dadurch aber die Hasenzahl ab und die nächste Fuchsgeneration muss hungern, wobei viele sterben. Dadurch geht es späteren Hasen besser usw. Die neue Hasenzahl  $H_{n+1}$  ergibt sich aus der alten  $H_n$ , vermehrt um die Neugeborenen, deren Zahl proportional zu  $H_n$  ist ( $c_1 \cdot H_n$ ), und vermindert um die Zahl der Gefressenen, deren Zahl von der Anzahl der Begegnungen zwischen Hasen und Füchsen abhängt ( $c_2 \cdot H_n \cdot F_n$ ):
 
$$H_{n+1} = H_n + c_1 \cdot H_n - c_2 \cdot H_n \cdot F_n$$

Ebenso erhält man die neue Fuchsanzahl  $F_{n+1}$  aus der alten  $F_n$ , vermindert um die Zahl der gestorbenen und vermehrt um die Zahl der neu geborenen, deren Zahl vom Futterzustand, also der Zahl der Begegnungen zwischen Hasen und Füchsen abhängt.

$$F_{n+1} = F_n - c_3 \cdot F_n - c_4 \cdot H_n \cdot F_n$$

Wir können die Zahl der Hasen auf der y-Achse auftragen, und die Zahl der Füchse auf der x-Achse. Einem bestimmten Zustand entspricht dann eine Kombination aus Hasen und Füchsen, die einen Punkt in der Hasen-Füchse-Ebene beschreibt: im *Phasenraum*.

- a: Stellen Sie die Entwicklung des Systems als Punktfolge in einem Diagramm dar.
- b: Experimentieren Sie mit den Konstanten des Systems und ermitteln Sie deren Einfluss. Welche Auswirkungen werden durch das Modell bestimmt, welche durch Einflüsse der Computerarithmetik. Wie kann man das feststellen?
- c: Führen Sie Möglichkeiten ein, die Konstanten am Bildschirm zu verändern.

3. Eine berühmte **Iterationsvorschrift** startet mit den Koordinaten eines Punktes  $C(a|b)$  in der Ebene. Aus diesen werden die Koordinaten der Folgepunkte  $X_n(x_n|y_n)$  nach der folgenden Vorschrift bestimmt:

im ersten Schritt:  $x_1 := a;$   $y_1 := b;$

weitere Schritte:  $x_{n+1} := x_n^2 - y_n^2 + a;$   $y_{n+1} := 2*x_n*y_n + b;$

Da es sich um eine Art Quadrierung mit nachfolgender Addition handelt, sollten Punkte in der Nähe des Ursprungs zu einer Punktfolge führen, die gegen den Ursprung konvergiert, während Punktfolgen, die in größerer Entfernung vom Ursprung starten, ins Unendliche laufen. Wir wollen jetzt die Punkte einer Umgebung des Ursprungs (z. B. innerhalb des Rechtecks mit den Eckpunkten  $(-2.0|1.1)$  und  $(0.6|-1.1)$ ) als Startwerte auffassen und dann jeweils die nächsten 100 Punkte der angegebenen Punktfolge berechnen. Liegt der letzte Punkt nahe beim Ursprung (z. B. näher als 2 Längeneinheiten), dann wollen wir den **Startpunkt** schwarz färben, sonst weiß. Die Rechnungen sind sehr zeitaufwendig und sollten deshalb nicht mit Hilfe der Timerkomponente gesteuert werden, sondern direkt in zwei geschachtelten Schleifen, in denen die a- bzw. die b-Koordinate des Startwerts in den angegebenen Grenzen

Für i von 0 bis RechtenBildrand tue
Für j von 0 bis UnterenBildrand tue
Berechne aus i und j die Koordinaten des Startpunktes
Führe das Iterationsverfahren wie angegeben durch
Färbe den Punkt (ij) je nach Ergebnis des Iterationsprozesses ein

Das Ergebnis ist die bekannte **Mandelbrotmenge**, das **Apfelmännchen**.

